

Universidade de São Paulo
Escola de Engenharia de São Carlos
Departamento de Engenharia Elétrica e de Computação

Análise do comando .MEAS do LTSpice

Fabício de Almeida Brito
Prof. José Marcos Alves

Novembro/ 2015

Índice

Análise do comando .MEAS		Pag.
1. Introdução		48
2. Tipos básicos de declaração		49
2.1. Medidas em apenas um ponto		49
2.2. Medidas em um intervalo		50
3. Exemplos de aplicação		52
3.1. Medidas em apenas um ponto		52
3.1.1. Medição num instante específico		53
3.1.2. Obtenção do instante no qual uma condição é satisfeita		56
3.1.3. Medição dependente do atendimento de uma condição		58
3.1.4. Cálculo de expressões envolvendo variáveis da simulação		60
3.2. Medidas em um intervalo		62
3.2.1. Medição computada sobre todo o intervalo de dados		63
3.2.2. Medição sobre intervalos diretamente declarados		64
3.2.3. Medição com intervalo definido pelo atendimento de condições		67
3.2.4. Obtenção do intervalo entre o atendimento de condições		72

Lista de Figuras

Fig.	3.1. Medidas em apenas um ponto	Pág.
1	Análise no domínio do tempo para um circuito RC	52
	3.1.1. Medição num instante específico	
2	Obtenção gráfica, por meio do uso de cursores, do valor de $V(\text{out})$, no instante 20 μs .	53
3	Resultado do comando “.meas” na determinação de $V(\text{out})$ em 20 μs . Obtido em Spice Error Log (Ctrl + L).	53
4	Obtenção gráfica, por meio do uso de cursores, do valor de $I(r) * V(\text{in}, \text{out})$ no instante 20 μs .	54
5	Resultado do comando “.meas” na determinação de $I(r) * V(\text{in}, \text{out})$ em 20 μs . Obtido em Spice Error Log (Ctrl + L).	54
6	Obtenção gráfica do valor da derivada de $V(\text{out})$ no instante 20 μs .	55
7	Resultado do comando “.meas” na determinação da derivada de $V(\text{out})$ em 20 μs . Obtido em Spice Error Log (Ctrl + L).	55
	3.1.2. Obtenção do instante no qual uma condição é satisfeita	
8	Obtenção gráfica dos instantes de tempo em que $V(\text{out}) = 0,5\text{V}$.	57
9	Resultado do comando “.meas” na determinação dos instantes de atendimento de $V(\text{out}) = 0,5\text{V}$. Obtido em Spice Error Log (Ctrl + L).	57
	3.1.3. Medição dependente do atendimento de uma condição	
10	Obtenção gráfica do valor de $I(R)$ quando $V(\text{out}) = 0,5\text{V}$.	58
11	Resultado do comando “.meas” na determinação de $I(R)$ quando $V(\text{out}) = 0,5\text{V}$. Obtido em Spice Error Log (Ctrl + L).	58
12	Obtenção gráfica do valor de $I(R) * V(\text{in}, \text{out})$ quando $V(\text{out}) = 0,5\text{V}$.	59
13	Resultado do comando “.meas” na determinação de $I(R) * V(\text{in}, \text{out})$ no último atendimento de $V(\text{out}) = 0,5\text{V}$. Obtido em Spice Error Log (Ctrl+L).	59
	3.1.4. Cálculo de expressões envolvendo variáveis da simulação	
14	Observação dos resultados res4 e res5, determinados pelo atendimento da condição $V(\text{out}) = 0,5\text{V}$.	60
15	Resultado do comando “.meas” na determinação de $3 * \text{res4} / \text{res5}$. Obtido em Spice Error Log (Ctrl+L).	60
16	Observação do resultado res6, determinado pelo segundo atendimento da condição $V(\text{out}) = 0,5\text{V}$.	61
17	Resultado do comando “.meas” na determinação de $\log_{10} \text{res6} $. Obtido em Spice Error Log (Ctrl+L).	61
	3.2. Medidas em um intervalo	
18	Filtro passa banda com frequência central de passagem em 1 kHz.	62
	3.2.1. Medição computada sobre todo o intervalo de dados	
19	Obtenção gráfica do máximo e do mínimo da magnitude de $V(\text{out})$.	63
20	Resultado do comando “.meas” na determinação do máximo e do mínimo da magnitude de $V(\text{out})$. Obtido em Spice Error Log (Ctrl+L).	64

Lista de Figuras (cont.)

Fig.	3.2. Medidas em um intervalo	Pág.
	3.2.2. Medição sobre intervalos diretamente declarados	
21	Obtenção gráfica do máximo e do mínimo da magnitude de V(out) no intervalo de 100 Hz a 1 kHz.	65
22	Resultado do comando “.meas” na determinação do máximo e do mínimo da magnitude de V(out) no intervalo de 100 Hz a 1 kHz. Obtido em Spice Error Log (Ctrl+L).	65
23	Obtenção gráfica do máximo da magnitude de V(out) no intervalo de 1 Hz a 1 kHz.	66
24	Resultado do comando “.meas” na determinação do máximo da magnitude de V(out) no intervalo de 1 Hz a 1 kHz. Obtido em Spice Error Log (Ctrl+L).	66
	3.2.3. Medição com intervalo definido pelo atendimento de condições	
25	Obtenção gráfica do mínimo da magnitude de V(out) no intervalo definido pelo primeiro atendimento das condições V(out) = 0,5 e V(out) = 2,0.	68
26	Resultado do comando “.meas” na determinação do mínimo da magnitude de V(out) no intervalo definido pelo primeiro atendimento das condições V(out) = 0,5 e V(out) = 2,0. Obtido em Spice Error Log (Ctrl+L).	68
27	Obtenção gráfica do valor máximo da magnitude de V(a), limitando a busca ao intervalo definido pelas passagens de V(out) pelo valor 0,5.	69
28	Resultado do comando “.meas” na determinação do máximo da magnitude de V(a), limitando a busca ao intervalo definido pelas passagens de V(out) pelo valor 0,5. Obtido em Spice Error Log (Ctrl+L).	69
29	Obtenção gráfica do valor máximo da magnitude de V(out), limitando a busca ao intervalo definido pela primeira passagens de V(out) pelo valor 0,5 e a frequência de 1 MHz.	70
30	Resultado do comando “.meas” na determinação do máximo da magnitude de V(out), limitando a busca ao intervalo definido pela primeira passagem de V(out) pelo valor 0,5 e a frequência de 1 MHz. Obtido em Spice Error Log (Ctrl+L).	70
31	Obtenção gráfica do valor máximo da magnitude de V(out), limitando a busca no intervalo definido pela primeira passagens de V(out) pelo valor 0,5 e a frequência de 1 kHz.	71
32	Resultado do comando “.meas” na determinação do máximo da magnitude de V(out), limitando a busca ao intervalo definido pela primeira passagem de V(out) pelo valor 0,5 e a frequência de 1 kHz. Obtido em Spice Error Log (Ctrl+L).	71
	3.2.4. Obtenção do intervalo entre o atendimento de condições	
33	Obtenção gráfica do intervalo de frequências limitado entre os atendimentos de $V(out) = \text{MAX}(V(out)) / \sqrt{2}$.	73
34	Resultado do comando “.meas” na determinação do intervalo limitado entre os atendimentos de $V(out) = \text{MAX}(V(out)) / \sqrt{2}$. Obtido em Spice Error Log (Ctrl+L).	73

1. Introdução

O comando “.meas” permite ao usuário realizar medições sobre o conjunto de dados de uma simulação, permitindo medidas como:

- Tempo de subida e de descida;
- Cálculo da média, do valor eficaz, do máximo, do mínimo e do valor pico-a-pico de um sinal;
- Medidas do tipo: Encontre o valor de X quando ocorrer Y;
- Cálculo da derivada ou da integral de um sinal;

Sua inserção é efetuada através da ferramenta “**SPICE directive**”, a qual pode ser acessada através da barra de ferramentas (último botão à direita), ou através do atalho no teclado “**s**”.

Os resultados do comando podem ser visualizados, apenas após o término da simulação, no arquivo “SPICE Error Log”, o qual pode ser acessado na barra de ferramentas em “View >>> SPICE Error Log” ou pelo atalho no teclado “**Ctrl + L**”.

2. Tipos básicos de declaração

Existem dois tipos básicos de declarações para a diretiva “.meas”: aquelas que se referem a apenas um ponto e aquelas que se referem a uma faixa de valores ao longo do eixo das abscissas (Eixo horizontal, por exemplo: eixo do tempo numa simulação “.tran”).

2.1. Medidas em apenas um ponto

A primeira versão, aquela que aponta para apenas um ponto, é usada para imprimir o valor de uma expressão em um ponto específico ou quando uma condição for atingida. A sintaxe para esse tipo de medição pode ser observada a seguir:

```
.MEAS [AC|DC|OP|TRAN|TF|NOISE] <NOME_VARIÁVEL>
+ [<FIND|DERIV|PARAM> <EXPR>]
+ [WHEN <EXPR> | AT = <EXPR>]
+ [TD = <VAL>]
+ [<RISE|FALL|CROSS> = [FIRST|<COUNT>|LAST]]1
```

O primeiro parâmetro, [AC|DC|OP|TRAN|TF|NOISE], especifica o tipo de simulação na qual a medição será efetuada. Tal informação é opcional e, caso omitida, o comando será executado para todos os tipos de simulação.

Em seguida, define-se o nome da variável (<nome_variável>) que receberá o resultado da medição. O nome da variável não pode ser omitido, nem repetido e pode ser utilizado na construção de uma expressão em outro comando “.meas” de forma a se recuperar um resultado salvo.

O parâmetro [<FIND|DERIV|PARAM> <expr>] define a expressão (<expr>) a ser utilizada, porém, com três diferentes abordagens. As duas primeiras são utilizadas para se retirar dados diretamente da simulação, com a diferença de que “find” apenas obtém o resultado da expressão, enquanto que “deriv” obtém o resultado da derivada da expressão. A última abordagem, “param”, é utilizada para retornar o resultado de expressões contendo apenas variáveis da simulação.

[WHEN <expr> | AT = <expr>] especifica o momento de atendimento de uma medição. Utiliza-se “at” para definir um ponto no eixo das abscissas no qual se deseja a medição. Por exemplo, no caso de uma simulação “.tran”, pode-se definir o instante de tempo exato no qual se deseja uma medição.

¹ Spice directives aceitam comandos multilinhas; para isso, deve-se inserir, no início das linhas adicionais, o símbolo +.

Utiliza-se “**when**” para especificar uma condição que, quando atendida, realizará uma medição. A condição especificada pode, durante a execução da simulação, ser atendida mais de uma vez. O padrão do LTspice é a execução da medição no primeiro atendimento da condição, no entanto, utilizando [**<RISE|FALL|CROSS> = [FIRST|<count>|LAST]**] pode-se alterar para outros momentos de atendimento da condição.

“**Rise**” é utilizado quando se deseja uma medição na parte crescente de um sinal ou expressão, “**fall**” para quando se deseja na parte decrescente e “**cross**” para um momento qualquer de atendimento da condição. Esses comandos devem ser seguidos de “**first**”, de “**last**” ou de um número inteiro positivo identificando o momento de atendimento da condição.

Por último, pode-se ainda acrescentar [**TD = <val>**], o qual representa um atraso, a partir do início da simulação, para que a medição seja efetuada.

2.2. Medidas em um intervalo

A segunda versão para o comando “**.meas**” refere-se a uma medição sobre um intervalo no eixo das abscissas. A sintaxe para esse caso é:

```
.MEAS [AC|DC|OP|TRAN|TF|NOISE] <NOME_VARIÁVEL>
+ [<AVG|MAX|MIN|PP|RMS|INTEG> <EXPR>]
+ [FROM <VAL1>]
+ [TRIG <EXPR_ESQ> [[VAL]=]<EXPR_DIR>]
+ [TD=<VAL3>]
+ [<RISE|FALL|CROSS> = [FIRST|<COUNT1>|LAST]]
+ [TO <VAL2>]
+ [TARG <EXPR_ESQ> [[VAL]=]<EXPR_DIR>]
+ [TD=<VAL4>]
+ [<RISE|FALL|CROSS> = [FIRST|<COUNT2>|LAST]]
```

A primeira parte do comando, [**AC|DC|OP|TRAN|TF|NOISE**] **<nome_variável>**, especifica o tipo de simulação na qual a medição será efetuada e o nome da variável que receberá o resultado de tal medição.

Então, especifica-se o tipo de operação e a expressão de análise: [**<AVG|MAX|MIN|PP|RMS|INTEG>** **<expr>**].

Os tipos de operação são divididos em:

- **AVG**: Computa o valor médio da expressão (**<expr>**);
- **MAX**: Determina o valor máximo de **<expr>**;
- **MIN**: Determina o valor mínimo de **<expr>**;
- **PP**: Determina o valor pico a pico de **<expr>**;

- RMS: Computa o valor eficaz de **<expr>**;
- INTEG: Computa a integral da expressão (**<expr>**);

O padrão do LTspice é realizar suas medições sobre o intervalo completo de simulação, no entanto, pode-se limitar tais medidas a um intervalo menor, por meio da determinação direta de seu início ou término ou por meio da declaração de condições que, quando atendidas, especificam o intervalo de medições.

Caso haja a necessidade, pode-se modificar apenas uma das extremidades do intervalo de medidas. Nesses casos, a extremidade não declarada é mantida tal como no padrão do programa, ou seja, igual a um dos limites do intervalo de simulação.

O primeiro método para a definição do intervalo de medidas é a declaração direta de seu início e término, os quais são realizados por meio de: **FROM <val1> TO <val2>**, sendo **<val1>** o início e **<val2>** o fim de tal intervalo.

O segundo método utiliza-se do atendimento de condições para estabelecer o intervalo de medidas. Esse método utiliza a seguinte sintaxe:

```
+ [TRIG <EXPR_ESQ> [ [VAL]=<EXPR_DIR> ]
+      [TD=<VAL3>]
+      [<RISE|FALL|CROSS> = [FIRST|<COUNT1>|LAST] ]
+ [TARG <EXPR_ESQ> [ [VAL]=<EXPR_DIR> ]
+      [TD=<VAL4>]
+      [<RISE|FALL|CROSS> = [FIRST|<COUNT2>|LAST] ]
```

A condição descrita após **TRIG** marca, com seu atendimento, o início do intervalo de medições e a descrita após **TARG** marca o fim de tal intervalo.

Tais condições podem ser construídas de duas formas: com sua igualdade na própria expressão, como por exemplo, **TRIG V(out) = 0.5**, ou por meio da separação dos termos da igualdade e da utilização de **[VAL]=<expr_dir>**, como por exemplo: **TRIG V(out) VAL = 0.5**.

O padrão do LTspice para o atendimento da condição descrita é para o primeiro momento no qual ela se tornar verdadeira, no entanto, pode-se alterar tal momento de atendimento utilizando: **[<RISE|FALL|CROSS> = [FIRST|<count>|LAST]]**.

“**Rise**” especifica que a condição somente se torna verdadeira nos momentos que o sinal ou a expressão de análise for crescente, “**fall**” nos momentos que a expressão de análise for decrescente e “**cross**” para qualquer momento de atendimento da condição. Esses comandos devem ser seguidos de “**first**”, “**last**” ou um número inteiro positivo, os quais identificam o momento desejado de atendimento da condição. Pode-se ainda acrescentar ao comando **[TD = <val>]**, o qual representa um atraso, a partir do início da simulação, no qual a condição definida não é atendida.

3. Exemplos de aplicação

Os exemplos abordados nesse documento buscam explorar todas as possíveis declarações do comando **“.meas”**, em medições pontuais e em medições num intervalo de valores do eixo das abscissas.

3.1. Medidas em apenas um ponto

A Figura 1, apresentada abaixo, apresenta um circuito constituído de uma fonte, um resistor de 1 k Ω e um capacitor de 10 nF. A fonte utilizada gera uma onda quadrada de 0 a 1 V com período igual a 200 μ s e tempo ativo de 100 μ s e a simulação realizada foi no domínio do tempo, de 0 a 200 μ s, a um passo de 1 ns.




Figura 1: Análise no domínio do tempo para um circuito RC

As medidas pontuais possíveis são divididas em quatro categorias: Medição num instante específico; Obtenção do instante no qual uma condição é satisfeita; Medição dependente do atendimento de uma condição; e Cálculo de expressões envolvendo variáveis da simulação.²

² As categorias citadas são também validas para outros tipos de simulação, tais como: simulações do ponto de operação (OP) e simulações no domínio da frequência (AC).

3.1.1. Medição num instante específico

Obtém o resultado de uma expressão ou de sua derivada em um instante de tempo específico, inserido no comando. A sintaxe do comando é:

```
.MEAS TRAN <NOME_VARIÁVEL> [FIND|DERIV] <EXPR> AT=<EXPR>
```

Exemplos:

```
.meas tran res1 find V(out) at=20u
```

Retorna, na variável res1, o valor de V(out) no instante de tempo $t = 20\mu\text{s}$.




Figura 2: Obtenção gráfica, por meio do uso de cursores, do valor de V(out), no instante $20\mu\text{s}$.

```
SPICE Error Log: C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.log

--- Expanded Netlist ---
* C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.asc
v in 0 pulse(0 1 0 1p 1p 100u 200u) rser=0
c out 0 10n
r in out 1k
.tran 0 200u 0 1n
.end

.OP point found by inspection.
res1: v(out)=0.864653 at 2e-005
```

Figura 3: Resultado do comando “.meas” na determinação de V(out) em $20\mu\text{s}$. Obtido em Spice Error Log (Ctrl + L).

`.meas tran res2 find I(r)*V(in,out) at =20u`

Retorna, na variável res2, o valor da expressão $I(r) * V(in,out)$ no instante de tempo $t = 20 \mu s$.




Figura 4: Obtenção gráfica, por meio do uso de cursores, do valor de $I(r) * V(in,out)$ no instante $20 \mu s$.

```

--- Expanded Netlist ---
* C:\Users\Fabricao\Desktop\LTspice .MEAS\Circuito_RC_TRAN.asc
v in 0 pulse(0 1 0 1p 1p 100u 200u) rser=0
c out 0 10n
r in out 1k
.tran 0 200u 0 1n
.end

.OP point found by inspection.
res2: i(r)*v(in,out)=1.83188e-005 at 2e-005

```

Figura 5: Resultado do comando “.meas” na determinação de $I(r) * V(in, out)$ em $20 \mu s$. Obtido em Spice Error Log (Ctrl + L).

`.meas tran res3 deriv V(out) at=20u`

Retorna, na variável res3, o valor da derivada de V(out) no instante de tempo $t = 20\mu\text{s}$.




Figura 6: Obtenção gráfica do valor da derivada de V(out) no instante $20\mu\text{s}$.

A janela 'SPICE Error Log: C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.log' exibe o seguinte conteúdo:

```

--- Expanded Netlist ---
* C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.asc
v in 0 pulse(0 1 0 1p 1p 100u 200u) rser=0
c out 0 10n
r in out 1k
.tran 0 200u 0 10n
.end

.OP point found by inspection.
res3: D(v(out))=13544.7 at 2e-005

```

Figura 7: Resultado do comando “.meas” na determinação da derivada de V(out) em $20\mu\text{s}$. Obtido em Spice Error Log (Ctrl + L).

3.1.2. Obtenção do instante no qual uma condição é satisfeita

Obtém o instante de tempo quando a condição estabelecida for atendida. Dependendo da declaração do comando, pode-se obter os diferentes momentos no qual a condição foi atendida. A sintaxe é:

```
.MEAS TRAN <NOME_VARIÁVEL> WHEN <EXPR>
+ [TD=<VAL>]
+ [<RISE | FALL | CROSS>= [<COUNT> | LAST ]
```

Exemplos: Todos os comandos descritos a seguir referem-se a uma mesma condição que, no intervalo de 0 a 200 μ s, é atendida mais de uma vez. Os resultados observados estão contidos nas Figuras 8 e 9.

```
.meas tran res4 when V(out)=0.5
```

Retorna, na variável res4, o tempo do primeiro momento em que $V(\text{out}) = 0,5$ for verdadeiro.

```
.meas tran res5 when V(out)=0.5 td=100u
```

Retorna, na variável res5, o tempo do primeiro atendimento da condição $V(\text{out}) = 0,5$, porém depois de um atraso de 100 μ s, isto é, o comando fica ativo apenas no intervalo de 100 a 200 μ s.

```
.meas tran res6 when V(out)=0.5 cross=2
```

Retorna, na variável res6, o instante de tempo do segundo atendimento da condição.

```
.meas tran res7 when V(out)=0.5 cross=last
```

Retorna, na variável res7, o tempo do último atendimento da condição.

```
.meas tran res8 when V(out)=0.5 rise=first
```

Retorna, na variável res8, o tempo do primeiro momento em que as duas condições forem verdadeiras: $V(\text{out}) = 0,5$ e sinal $V(\text{out})$ crescente.

```
.meas tran res9 when V(out)=0.5 fall=last
```

Retorna, na variável res9, o tempo do último momento em que as duas condições forem verdadeiras: $V(\text{out}) = 0,5$ e sinal $V(\text{out})$ decrescente.




Figura 8: Obtenção gráfica dos instantes de tempo em que $V(\text{out}) = 0,5V$.

```

SPICE Error Log: C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.log

* C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.asc
v in 0 pulse(0 1 0 lp lp 100u 200u) rser=0
c out 0 10n
r in out 1k
.tran 0 200u 0 10n
.end

.OP point found by inspection.

res4: v(out)=0.5 AT 6.93253e-006
res5: v(out)=0.5 AT 0.000106929
res6: v(out)=0.5 AT 0.000106929
res7: v(out)=0.5 AT 0.000106929
res8: v(out)=0.5 AT 6.93253e-006
res9: v(out)=0.5 AT 0.000106929

```

Figura 9: Resultado do comando “.meas” na determinação dos instantes de atendimento de $V(\text{out}) = 0,5V$. Obtido em Spice Error Log (Ctrl + L).

3.1.3. Medição dependente do atendimento de uma condição

Obtém o resultado de uma expressão ou de sua derivada no instante que a condição estabelecida for atendida. Dependendo da declaração do comando, pode-se obter diferentes resultados, os quais refletem nos diferentes momentos no qual a condição foi atendida. A sintaxe para esse caso é:

```
.MEAS TRAN <NOME_VARIÁVEL> FIND <EXPR> WHEN <EXPR>
+ [TD=<VAL>]
+ [<RISE | FALL | CROSS>= [<COUNT> | LAST ]
```

Exemplos:

```
.meas tran res10 find I(r) when V(out)=0.5
```

Retorna, na variável res10, o valor de I(r) quando a condição V(out) = 0,5 for satisfeita pela primeira vez.




Figura 10: Obtenção gráfica do valor de I(R) quando V(out) = 0,5V.

```
SPICE Error Log: C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.log
--- Expanded Netlist ---
* C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.asc
v in 0 pulse(0 1 0 1p 1p 100u 200u) rser=0
c out 0 10n
r in out 1k
.tran 0 200u 0 1n
.end

.OP point found by inspection.
res10: i(r)=0.0005 at 6.93213e-006
```

Figura 11: Resultado do comando ".meas" na determinação de I(R) quando V(out) = 0,5V. Obtido em Spice Error Log (Ctrl + L).

```
.meas tran res11 find I(r)*V(in,out)
+ when v(out)=0.5 cross=last
```

Retorna, na variável res11, o valor da expressão $I(r) * V(in,out)$ quando a condição $V(out) = 0,5$ for satisfeita pela última vez.




Figura 12: Obtenção gráfica do valor de $I(R) * V(in, out)$ quando $V(out) = 0,5V$.

```
--- Expanded Netlist ---
* C:\Users\Fabricao\Desktop\LTspice .MEAS\Circuito_RC_TRAN.asc
v in 0 pulse(0 1 0 1p 1p 100u 200u) rser=0
c out 0 10n
r in out 1k
.tran 0 200u 0 1n
.end

.OP point found by inspection.

res11: i(r)*v(in,out)=0.00025 at 0.000106932
```

Figura 13: Resultado do comando “.meas” na determinação de $I(R) * V(in, out)$ no último atendimento de $V(out) = 0,5V$. Obtido em Spice Error Log (Ctrl+L).

3.1.4. Cálculo de expressões envolvendo variáveis da simulação

Esse tipo de declaração do comando “.meas” obtém o resultado de uma expressão construída apenas com variáveis da simulação. Sendo assim, não se obtém novos dados, apenas manipula-se os existentes. A sintaxe é:

.MEAS [TRAN] <NOME_VARIÁVEL> PARAM <EXPR>

Exemplos:

.meas tran res12 param 3*res4/res5

Retorna, na variável res12, o valor da expressão $3 * \text{res4} / \text{res5}$ apenas se uma simulação “.tran” tiver sido realizada.




Figura 14: Os resultados res4 e res5 são instantes de tempo determinados pelo atendimento da condição $V(\text{out}) = 0,5\text{V}$.

```

SPICE Error Log: C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.log
--- Expanded Netlist ---
* C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.asc
v in 0 pulse(0 1 0 1p 1p 100u 200u) rser=0
c out 0 10n
r in out 1k
.tran 0 200u 0 1n
.end

.OP point found by inspection.

res4: v(out)=0.5 AT 6.93213e-006
res5: v(out)=0.5 AT 0.000106932
res12: 3*res4/res5=0.194482
  
```

Figura 15: Resultado do comando “.meas” na determinação de $3 * \text{res4} / \text{res5}$. Obtido em Spice Error Log (Ctrl+L).

`.meas res13 param log10(abs(res6))`

Retorna, na variável res13, o valor da expressão $\log_{10} |res6|$, não importando o tipo de simulação realizada.




Figura 16: O resultado res6 representa o instante de tempo determinado pelo atendimento da condição $V(out) = 0,5V$.

```

SPICE Error Log: C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.log
--- Expanded Netlist ---
* C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_TRAN.asc
v in 0 pulse(0 1 0 1p 1p 100u 200u) rser=0
c out 0 10n
r in out 1k
.tran 0 200u 0 1n
.end


.OP point found by inspection.

res6: v(out)=0.5 AT 0.000106932
res13: log10(abs(res6))=-3.97089
  
```

Figura 17: Resultado do comando “.meas” na determinação de $\log_{10}|res6|$. Obtido em Spice Error Log (Ctrl+L).

3.2. Medidas em um intervalo

A Figura 17 apresenta um filtro passa banda ativo, projetado para um ganho de 2V/V (6,02 dB) na frequência 1 kHz, sua frequência central de passagem. A simulação realizada foi no domínio da frequência, de 1Hz a 1 MHz, com 1000 pontos por década.



; SIMULAÇÃO REALIZADA
.ac dec 1000 1 1meg

Figura 18: Filtro passa banda com frequência central de passagem em 1 kHz.

As medidas possíveis são divididas em quatro categorias: Medição computada sobre todo o intervalo de dados; Medição sobre intervalos diretamente declarados; Medição com intervalo definido pelo atendimento de condições; e Obtenção do intervalo entre o atendimento de condições.³

³ As categorias citadas são também válidas para outros tipos de simulação, tais como: simulações do ponto de operação (OP) e simulações no domínio do tempo (TRAN).

3.2.1. Medição computada sobre todo o intervalo de dados

Obtém a média, o máximo, o mínimo, o valor pico-a-pico, o valor eficaz ou a integral de uma expressão computados sobre todo o intervalo de simulação. A sintaxe para esse caso é:

```
.MEAS AC <NOME_VARIÁVEL>  
+ [<AVG|MAX|MIN|PP|RMS|INTEG> <EXPR>]
```

Exemplos:

Os comandos descritos a seguir referem-se à uma mesma expressão de análise, diferindo apenas no fato de um computar o máximo e o outro computar o mínimo da expressão. Sendo assim, seus resultados foram agrupados e podem ser observados nas Figuras 19 e 20.

```
.meas ac res1 max mag(V(out))
```

Retorna, na variável res1, o valor máximo do sinal $\text{mag}(V(\text{out}))^4$ no intervalo completo de simulação.

```
.meas ac res2 min mag(V(out))
```

Retorna, na variável res2, o valor mínimo do sinal $\text{mag}(V(\text{out}))^4$ no intervalo completo de simulação.




Figura 19: Obtenção gráfica do máximo e do mínimo da magnitude de $V(\text{out})$.

⁴ Computa a magnitude do sinal $V(\text{out})$

```

SPICE Error Log: C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_AC.log
res1: MAX(mag(v(out)))=(6.27899dB,0°) FROM 1 TO 1e+006
res2: MIN(mag(v(out)))=(-53.9443dB,0°) FROM 1 TO 1e+006

Date: Mon Nov 23 11:27:51 2015
Total elapsed time: 0.160 seconds.

```

Figura 20: Resultado do comando “.meas” na determinação do máximo e do mínimo da magnitude de V(out). Obtido em Spice Error Log (Ctrl+L).

3.2.2. Medição sobre intervalos diretamente declarados

Computa a média, o máximo, o mínimo, o valor pico-a-pico, o valor eficaz ou a integral de uma expressão sobre um intervalo de análise definido diretamente na declaração do comando “.meas”.

Nesse tipo de declaração, pode-se alterar as duas extremidades do intervalo de análise ou apenas uma. Nos casos de se modificar apenas uma das extremidades, a extremidade não declarada é mantida tal como no padrão do programa, ou seja, igual a um dos limites do intervalo de simulação. A sintaxe é:

```

.MEAS AC <NOME_VARIÁVEL>
+ [<AVG | MAX | MIN | PP | RMS | INTEG> <EXPR>]
+ [FROM <VAL1>]
+ [TO <VAL2>]

```

Exemplos:

Os dois comandos seguintes referem-se à uma mesma expressão de análise. Sendo assim, seus resultados foram agrupados e podem ser observados nas Figuras 21 e 22.

```
.meas ac res3 min mag(V(out)) from 100 to 1k
```

Retorna, na variável res3, o valor mínimo da magnitude de V(out) no intervalo de 100Hz a 1 kHz.

```
.meas ac res4 max mag(V(out)) from 100 to 1k
```

Retorna, na variável res4, o valor máximo da magnitude de V(out) no intervalo de 100Hz a 1 kHz.




Figura 21: Obtenção gráfica do máximo e do mínimo da magnitude de $V(out)$ no intervalo de 100 Hz a 1 kHz.




Figura 22: Resultado do comando ".meas" na determinação do máximo e do mínimo da magnitude de $V(out)$ no intervalo de 100 Hz a 1 kHz. Obtido em Spice Error Log (Ctrl+L).

`.meas ac res5 max mag(V(out)) to 1k`

Retorna, na variável `res5`, o valor máximo da magnitude de `V(out)` no intervalo terminado em 1 kHz. Como a simulação realizada é de 1 Hz a 1 MHz, o intervalo de análise da expressão `mag(V(out))` é de 1 Hz a 1 kHz.




Figura 23: Obtenção gráfica do máximo da magnitude de `V(out)` no intervalo de 1 Hz a 1 kHz.

```

SPICE Error Log: C:\Users\Fabrcio\Desktop\LTspice .MEAS\Circuito_RC_AC.log
res5: MAX(mag(v(out)))=(6.277dB,0°) FROM 1 TO 1000

Date: Mon Nov 23 11:53:47 2015
Total elapsed time: 0.193 seconds.

```

Figura 24: Resultado do comando `.meas` na determinação do máximo da magnitude de `V(out)` no intervalo de 1 Hz a 1 kHz. Obtido em Spice Error Log (Ctrl+L).

3.2.3. Medição com intervalo definido pelo atendimento de condições

Computa a média, o máximo, o mínimo, o valor pico-a-pico, o valor eficaz ou a integral de uma expressão sobre um intervalo de análise definido pelo atendimento de condições.

Nesse tipo de declaração, pode-se definir condições de atendimento para as duas extremidades do intervalo de análise ou apenas uma extremidade. Nos casos de se modificar apenas uma das extremidades, a extremidade não declarada é mantida tal como no padrão do programa, ou seja, igual a um dos limites do intervalo de simulação.

Além disso, a definição do intervalo de análise pode ser construída com a combinação de uma declaração direta e uma declaração por meio do atendimento de condições.

A sintaxe do comando é:

```
.MEAS AC <NOME_VARIÁVEL>
+ [<AVG|MAX|MIN|PP|RMS|INTEG> <EXPR>]
+ [FROM <VAL1>]
+ [TRIG <EXPR_ESQ> [[VAL]=]<EXPR_DIR>]
+ [TD=<VAL3>]
+ [<RISE|FALL|CROSS> = [FIRST|<COUNT1>|LAST]]
+ [TO <VAL2>]
+ [TARG <EXPR_ESQ> [[VAL]=]<EXPR_DIR>]
+ [TD=<VAL4>]
+ [<RISE|FALL|CROSS> = [FIRST|<COUNT2>|LAST]]
```

Exemplos:

```
.meas ac res6 min mag(V(out))
+ trig mag(V(out)) val=0.5
+ targ mag(V(out)) val=2
```

Computa o mínimo da magnitude do sinal V(out) utilizando o intervalo definido entre o primeiro momento que a magnitude assume 0,5 e o primeiro momento que a magnitude assume 2.




Figura 25: Obtenção gráfica do mínimo da magnitude de V(out) no intervalo definido pelas condições V(out) = 0,5 e V(out) = 2,0.




Figura 26: Resultado do comando ".meas" na determinação do mínimo da magnitude de V(out) no intervalo definido pelo primeiro atendimento das condições V(out) = 0,5 e V(out) = 2,0. Obtido em Spice Error Log (Ctrl+L).

```
.meas ac res7 max mag(V(a))
+ trig mag(V(out)) val=0.5 cross=first
+ targ mag(V(out)) val=0.5 cross=last
```

Computa o máximo da magnitude do sinal V(a) utilizando o intervalo definido entre os momentos que a magnitude de V(out) assume 0,5, isto é, entre o primeiro e o último momento que V(out) = 0,5.




Figura 27: Obtenção gráfica do valor máximo da magnitude de V(a), limitando a busca ao intervalo definido pelas passagens de V(out) pelo valor 0,5.




Figura 28: Resultado do comando “.meas” na determinação do máximo da magnitude de V(a), limitando a busca ao intervalo definido pelas passagens de V(out) pelo valor 0,5. Obtido em Spice Error Log (Ctrl+L).

```
.meas ac res8 max mag(V(out))
+ trig mag(V(out)) val=0.5
```

Computa o máximo da magnitude do sinal V(out) utilizando o intervalo definido entre o primeiro momento que a magnitude de V(out) assume 0,5 e o limite superior da simulação, isto é, o intervalo de análise vai até 1 MHz.




Figura 29: Obtenção gráfica do valor máximo da magnitude de V(out), limitando a busca ao intervalo definido pela primeira passagem de V(out) pelo valor 0,5 e a frequência de 1 MHz.




Figura 30: Resultado do comando “.meas” na determinação do máximo da magnitude de V(out), limitando a busca ao intervalo definido pela primeira passagem de V(out) pelo valor 0,5 e a frequência de 1 MHz. Obtido em Spice Error Log (Ctrl+L).

```
.meas ac res8 max mag(V(out))
+ trig mag(V(out)) val=0.5
+ to 1k
```

Computa o máximo da magnitude do sinal V(out) utilizando o intervalo definido entre o primeiro momento que a magnitude de V(out) assume 0,5 e a frequência 1 kHz.




Figura 31: Obtenção gráfica do valor máximo da magnitude de V(out), limitando a busca no intervalo definido pela primeira passagem de V(out) pelo valor 0,5 e a frequência de 1 kHz.




Figura 32: Resultado do comando“.meas” na determinação do máximo da magnitude de V(out), limitando a busca ao intervalo definido pela primeira passagem de V(out) pelo valor 0,5 e a frequência de 1 kHz. Obtido em Spice Error Log (Ctrl+L).

3.2.4. Obtenção do intervalo entre o atendimento de

Esse tipo de declaração retorna o intervalo entre o atendimento de condições. No entanto, pode-se definir apenas uma condição ou duas, dependendo de como se desejar a construção do intervalo.

Nos casos de se modificar apenas uma das extremidades, a extremidade não declarada é mantida tal como no padrão do programa, ou seja, igual a um dos limites do intervalo de simulação.

Além disso, a definição do intervalo pode ser construída com a combinação de uma declaração direta e uma declaração por meio do atendimento de condições.

A sintaxe para esse caso é:

```
.MEAS AC <NOME_VARIÁVEL>
+ [FROM <VAL1>]
+ [TRIG <EXPR_ESQ> [[VAL]=]<EXPR_DIR>]
+ [TD=<VAL3>]
+ [<RISE|FALL|CROSS> = [FIRST|<COUNT1>|LAST]]
+ [TO <VAL2>]
+ [TARG <EXPR_ESQ> [[VAL]=]<EXPR_DIR>]
+ [TD=<VAL4>]
+ [<RISE|FALL|CROSS> = [FIRST|<COUNT2>|LAST]]
```

Exemplo:

```
.meas ac Acorte max mag(v(out))/sqrt(2)
.meas ac res10
+ trig mag(V(out)) val=Acorte cross=first
+ targ mag(V(out))=Acorte cross=last5
```

Realiza duas medições sobre o sinal: magnitude de V(out). A primeira, determina o valor da magnitude na sua frequência de corte, calculado tomando o valor máximo da magnitude de V(out) e dividindo por $\sqrt{2}$. A segunda medição determina o intervalo entre os momentos em que a magnitude de V(out) assume o valor Acorte, sendo assim, por definição, a variável res10 pode ser entendida como sendo a largura de banda calculada para o filtro em estudo, afinal, $BW = Fca - Fcb$, sendo Fca a frequência de corte superior e Fcb a frequência de corte inferior.

⁵ Observar que a declaração da expressão de análise pode ser realizada de duas maneiras: uma por meio da separação do sinal de análise e o valor que ele deve assumir (**mag(V(out)) val = Acorte**) e outra por meio da declaração direta da expressão (**mag(V(out)) = Acorte**).




Figura 33: Obtenção gráfica do intervalo de frequências limitado entre os atendimentos de $V(\text{out}) = \text{MAX}(V(\text{out})) / \sqrt{2}$.




Figura 34: Resultado do comando ".meas" na determinação do intervalo limitado entre os atendimentos de $V(\text{out}) = \text{MAX}(V(\text{out})) / \sqrt{2}$. Obtido em Spice Error Log (Ctrl+L).